MAT3243.001 Fall '09 Midterm 1

```
> with(linalg):
Warning, the protected names norm and trace have been redefined and
unprotected
```

```
> with(plots):
Warning, the name changecoords has been redefined
```

#1

```
> f:=sqrt(4*x^2+y^2);
```

$$f := \sqrt{4\,x^2 + y^2}$$

```
> contourplot(f,x=-3..3,y=-3..3,scaling=constrained);
```




```
> plot3d(f,x=-3..3,y=-3..3,axes=boxed,style=patchcontour,scaling=c
onstrained);
```



#2 (a) different directional limits, so the limit does not exist.

```
> f:=(x*y-y^2)/(x^2+y^2);
  subs({x=0},f);
```

```
subs({y=0},f);
```

$$f := \frac{x\,y - y^2}{x^2 + y^2}$$

$$-1$$

$$0$$

(b) factor, cancel, the limit is 0.

```
> (x^6-x^2*y^4)/(x^2+y^2);
  factor(%);
  subs({x=0,y=0},%);
```

$$\frac{x^6 - x^2\,y^4}{x^2 + y^2}$$

$$(x - y)\,(x + y)\,x^2$$

$$0$$

#3

```
> T:=98-x^2*y;
```

$$T := 98 - x^2\,y$$

```
> theta:=-Pi/6;
```

$$\theta := -\frac{\pi}{6}$$

Direction is given by the unit vector

```
> u:=[cos(theta),sin(theta)];
```

$$u := \left[\frac{\sqrt{3}}{2}, \frac{-1}{2}\right]$$

Directional derivative = grad T evaluated at [1,2] dot u
Rate of change = directional derivative * speed (degrees/time = degrees/distance * distance/time)

```
> grad(T,[x,y]);
  subs({x=1,y=2},%);
  dotprod(%,u);
  %*4;
  evalf(%);
```

$$[-2\,x\,y, -x^2]$$

$$[-4, -1]$$

$$\frac{1}{2} - 2\sqrt{3}$$

$$2 - 8\sqrt{3}$$

$$-11.85640646$$

#4

```
> ss:={x=0,y=0};
```

$$ss := \{x = 0, y = 0\}$$

```
> f:=exp(1+x+y^2);
  ff:=subs(ss,f);
```

$$f := e^{(1+x+y^2)}$$

$$ff := e$$

```
> jacobian([f],[x,y]);
  Df:=subs(ss,%);
```

$$[e^{(1+x+y^2)} \quad 2\,y\,e^{(1+x+y^2)}]$$

$$Df := [e \quad 0]$$

```
> hessian(f,[x,y]);
  Hf:=subs(ss,%);
```

$$\begin{bmatrix} e^{(1+x+y^2)} & 2\,y\,e^{(1+x+y^2)} \\ 2\,y\,e^{(1+x+y^2)} & 2\,e^{(1+x+y^2)} + 4\,y^2\,e^{(1+x+y^2)} \end{bmatrix}$$

$$Hf := \begin{bmatrix} e & 0 \\ 0 & 2\,e \end{bmatrix}$$

Displacement vector

```
> h:=[x,y];
```

$$h := [x, y]$$

Quadratic approximation formula

```
> ff+Df &* h+(1/2)*transpose(h) &* Hf &* h;
  evalm(%); evalf(%);
```

$$e + (Df \,\&*\, [x, y]) + \left(\left(\left(\frac{1}{2}\,\mathrm{transpose}([x, y])\right) \&*\, Hf\right) \&*\, [x, y]\right)$$

$$\left[e\,x + e + \frac{1}{2}\,e\,x^2 + y^2\,e\right]$$

$$[2.718281828\,x + 2.718281828 + 1.359140914\,x^2 + 2.718281828\,y^2]$$

Check: e^(1+x+y^2)=e*e^(x+y^2). Take the quadratic approximation to e^z and plug in x+y^2. Toss terms of degree > 2.

```
> convert(series(exp(z),z,3),polynom);
  exp(1)*subs(z=x+y^2,%): expand(%);
```

$$1 + z + \frac{1}{2}\,z^2$$

$$e + e\,x + y^2\,e + \frac{1}{2}\,e\,x^2 + e\,x\,y^2 + \frac{1}{2}\,e\,y^4$$

#5

```
> dd:=2*x^2+y^2-y+3;
```

$$dd := 2\,x^2 + y^2 - y + 3$$

Look for interior critical points by setting the gradient to 0 and solving

```
> grad(dd,[x,y]);
  solve(convert(%,set),{x,y}):
  crit:=subs(%,[x,y]);
```

$$[4\,x, 2\,y - 1]$$

$$crit := \left[0, \frac{1}{2}\right]$$

Look for boundary critical points by parametrizing the boundary, plugging in, taking the derivative, setting it to 0 and solving for t.
Construct a table of the corresponding points and values, pick the largest and the smallest values.
Note: solution t=-pi/2 is missed by Maple's solve, so I added it manually

```
> subs({x=cos(t),y=sin(t)},dd);
  diff(%,t);
  solve(%,t);
  %,-Pi/2;
  [crit,op(map(tt->[cos(tt),sin(tt)],[%]))]:
  [op(%)];
  map(pp->subs({x=pp[1],y=pp[2]},dd),%);
  evalf(%);
  values:=op(%):
```

$$2\cos(t)^2 + \sin(t)^2 - \sin(t) + 3$$

$$-2\cos(t)\sin(t) - \cos(t)$$

$$\frac{\pi}{2}, -\frac{\pi}{6}, -\frac{5\pi}{6}$$

$$\frac{\pi}{2}, -\frac{\pi}{6}, -\frac{5\pi}{6}, -\frac{\pi}{2}$$

$$\left[\left[0, \frac{1}{2}\right], [0, 1], \left[\frac{\sqrt{3}}{2}, \frac{-1}{2}\right], \left[-\frac{\sqrt{3}}{2}, \frac{-1}{2}\right], [0, -1]\right]$$

$$\left[\frac{11}{4}, 3, \frac{21}{4}, \frac{21}{4}, 5\right]$$

$$[2.750000000, 3., 5.250000000, 5.250000000, 5.]$$

```
> max(values);
  min(values);
```

$$5.250000000$$

$$2.750000000$$

```
>
```